```
In[1]:= ClearAll["Global`*"]

In[2]:= (* Bézier cubics throughout *)

In[3]:= $Assumptions = t ∈ Reals && t ≥ 0 && t ≤ 1 && z ∈ Reals && z > 0 && z < 1;

In[4]:= KnotsFromCoeffs[{c0_, c1_, c2_, c3_}] =
          {c0, c0 + c1 / 3, c0 + (2 c1 + c2) / 3, c0 + c1 + c2 + c3};
        CoeffsFromKnots[{z0_, z1_, z2_, z3_}] =
          {z0, -3 z0 + 3 z1, 3 z0 - 6 z1 + 3 z2, -z0 + 3 z1 - 3 z2 + z3};

In[6]:= (* Simple check *)
        KnotsFromCoeffs[CoeffsFromKnots[{z0, z1, z2, z3}]] // Simplify

Out[6]= {z0, z1, z2, z3}

In[7]:= (* Simple check *)
        CoeffsFromKnots[KnotsFromCoeffs[{c0, c1, c2, c3}]] // Simplify

Out[7]= {c0, c1, c2, c3}

In[8]:= tPowers = {1, t, t^2, t^3};

In[9]:= (* Curve starts at (1,0) *)
        x0 = 1; y0 = 0;
        (* Curve ends Angle widdershins away from that *)
        x3 = Cos[Angle]; y3 = Sin[Angle];
        (* Middle knots distance z away from from ends, in required direction *)
        x1 = 1; y1 = z;
        x2 = x3 + z Sin[Angle]; y2 = y3 - z Cos[Angle];
        x = Simplify[tPowers.CoeffsFromKnots[{x0, x1, x2, x3}]];
        y = Simplify[tPowers.CoeffsFromKnots[{y0, y1, y2, y3}]];

In[15]:= Block[{Angle = π / 2, z = 0.552}, ParametricPlot[{x, y}, {t, 0, 1}]]
```

Out[15]=



In[16]:=
```
(*
    (* I prefer to do this as the minimisation of an integral,
but that is difficult. *)
    partialImperfectness=FullSimplify[(x^2+y^2-1)^2 D[ArcTan[y/x],t]];
    partialImperfectnessZ=FullSimplify[D[partialImperfectness,z]];
    imperfectnessZ=Integrate[partialImperfectnessZ,{t,0,1}];
    Solve[imperfectnessZ==0,z]
*)
```

In[17]:= `Solve[(x /. {t → 1 / 2}) == Cos[Angle / 2], z]`

Out[17]= $\left\{\left\{z \to \frac{4}{3}\left(-1 + 2\,\mathrm{Cos}\left[\frac{\mathrm{Angle}}{2}\right] - \mathrm{Cos}[\mathrm{Angle}]\right)\mathrm{Csc}[\mathrm{Angle}]\right\}\right\}$

In[18]:= `z = FullSimplify[z /. %[[1]]]`

Out[18]= $\frac{4}{3}\,\mathrm{Tan}\left[\frac{\mathrm{Angle}}{4}\right]$

In[19]:= `Plot[z /. {Angle → a π / 180}, {a, 0, 90}]`

Out[19]=



In[20]:= `(* How different from a straight line? *)`

In[21]:= `Plot[(z /. {Angle → a π / 180}) - a/90 (z /. {Angle → π / 2}), {a, 0, 90}]`

Out[21]=



In[22]:= `(* So slightly different. *)`

In[23]:= **Block**[{**Angle** = π / 6},
ParametricPlot[{**x**, **y**}, {**t**, 0, 1}, **AspectRatio** → $\frac{\textbf{Sin[Angle]}}{\textbf{1 - Cos[Angle]}}$]]

Out[23]=


In[24]:= **error = FullSimplify**[$x^2 + y^2 - 1$]

Out[24]= $16\,t^2\,(1 - 3\,t + 2\,t^2)^2\,\text{Sin}\left[\frac{\text{Angle}}{4}\right]^4\,\text{Tan}\left[\frac{\text{Angle}}{4}\right]^2$

In[25]:= **Solve[D[error, t] == 0 && t > 0 && t < 1 && t ≠ 1 / 2, t]**

Out[25]= $\left\{\left\{t \to \frac{1}{6}\left(3 - \sqrt{3}\right)\right\}, \left\{t \to \frac{1}{6}\left(3 + \sqrt{3}\right)\right\}\right\}$

In[26]:= **error = FullSimplify[error /. %[[1]]]**

Out[26]= $\frac{4}{27}\,\text{Sin}\left[\frac{\text{Angle}}{4}\right]^4\,\text{Tan}\left[\frac{\text{Angle}}{4}\right]^2$

In[27]:=
```
LogPlot[(Sqrt[error + 1] - 1) /. {Angle → aDeg Degree}, {aDeg, 0, 90}]
```

Out[27]=



In[28]:=
```
(* In basis points. At my printer's 1200 dpi
 an error of 1bp is one pixel in a radius of 8⅓". *)
Table[{aDeg, N[10 000 (Sqrt[error + 1] - 1) /. {Angle → aDeg Degree}]},
  {aDeg, 90, 5, -5}] // TableForm
```

Out[28]//TableForm=

90	2.7253
85	1.9328
80	1.34269
75	0.911183
70	0.602095
65	0.385857
60	0.238644
55	0.14156
50	0.0798954
45	0.0424553
40	0.0209404
35	0.00939746
30	0.00372662
25	0.00124801
20	0.000327155
15	0.0000582263
10	5.11176×10^{-6}
5	7.98717×10^{-8}

In[29]:=
```
Series[error, {Angle, 0, 6}]
```

Out[29]= $\frac{\text{Angle}^6}{27\,648} + O[\text{Angle}]^7$

In[30]:=
```
Series[error /. {Angle → aDeg π / 180}, {aDeg, 0, 6}]
```

Out[30]= $\frac{\pi^6\ \text{aDeg}^6}{940\,369\,969\,152\,000\,000} + O[\text{aDeg}]^7$

In[31]:=
```
FactorInteger[940 369 969 152 000 000]
```

Out[31]= {{2, 22}, {3, 15}, {5, 6}}

In[32]:=
```
Plot[((Sqrt[error + 1] - 1) /. {Angle → aDeg π / 180}) / (π^6 aDeg^6 / (2^23 3^15 5^6)), {aDeg, 5, 90}]
```

Out[32]=



In[33]:=
```
Plot[((Sqrt[error + 1] - 1) /. {Angle → aDeg π / 180}) / (π^6 aDeg^6 / (2^23 3^15 5^6)), {aDeg, 0, 5}]
```

Out[33]=



In[34]:=
```
Plot[((Sqrt[error + 1] - 1) /. {Angle → aDeg π / 180}) / (π^6 aDeg^6 / (2^23 3^15 5^6)),
  {aDeg, 0, 5}, WorkingPrecision → 30]
```

Out[34]=